Math 2650 - Linear Differential Equations

A. J. Meir

Examples and Things to Come

An ordinary differential equation is an equation which involves an unknown function and its derivatives with respect to one independent variable. (A partial differential equation involves an unknown function and its partial derivatives with respect to several independent variables.)

The pendulum

$$m \frac{d^2}{dt^2}\phi + \frac{c}{l}\frac{d}{dt}\phi + \frac{m\,g}{l}\sin(\phi) = F(t)$$

$$m \frac{d^2}{dt^2}\phi + \frac{c}{l}\frac{d}{dt}\phi + \frac{m\,g}{l}\phi = F(t)$$

Population growth

$$\frac{d}{dt}p = \text{birth rate - death rate}$$

$$= b\,p - d\,p$$

Radioactive decay

$$\frac{d}{dt} q = -k\, q$$

In general an nth order o.d.e. is an equation of the form

$$F\left(t, x, \frac{dx}{dt}, \frac{d^2 x}{dt^2}, \ldots, \frac{d^n x}{dt^n}\right) = 0$$

and a general first order o.d.e. is an equation of the form

$$\frac{dx}{dt} = f(t, x)$$

A solution of a differential equation is a differentiable function defined on some interval I which satisfies the equation (the o.d.e.) at every point in I. The solution may given explicitly or implicitly. That is this function may be defined explicitly or implicitly.

An explicit solution ϕ is a solution of the form

$\phi(t)$ = expression involving only t (the independent variable)

and an implicit solution Φ is of the form

$$\Phi(t, x) = 0$$

(an expression involving both x and t, the dependent and independent variables) it implicitly defines the function $x(t)$ (the solution of the differential equation).

```
> restart:with(plottools):with(plots):
Warning, the name changecoords has been redefined
```

Consider the radioactive decay problem $\frac{d}{dt}$ q = -k q

```
> eq1:=diff(q(t),t)=-k*q(t);
```

$$eq1 := \frac{d}{dt} q(t) = -k\, q(t)$$

```
> dsolve(eq1,q(t));
```

$$q(t) = \_C1\, \mathbf{e}^{(-k\, t)}$$

```
> ic1:=q(0)=q[0];
```

$$ic1 := q(0) = q_0$$

```
> dsolve({eq1,ic1},q(t));
```

$$q(t) = q_0\, \mathbf{e}^{(-k\, t)}$$

```
> soln1:=rhs(%);
```

$$soln1 := q_0\, \mathbf{e}^{(-k\, t)}$$

```
> q[0]:=2;k:=0.1;
```

$$q_0 := 2$$

$$k := 0.1$$

```
> soln1;
```

$$2\, \mathbf{e}^{(-0.1\, t)}$$

```
> plot(soln1,t=0..100,thickness=2);
```



Consider the population growth $\frac{d}{dt}$ p = b p - d p

```
> eq2:=diff(p(t),t)=b*p(t)-d*p(t);
```

$$eq2 := \frac{d}{dt} p(t) = b\, p(t) - d\, p(t)$$

```
> dsolve(eq2,p(t));
```

$$p(t) = \_C1\, \mathbf{e}^{((b - d)\, t)}$$

```
> ic2:=p(0)=p[0];
```

$$ic2 := p(0) = p_0$$

```
> dsolve({eq2,ic2},p(t));
```

$$p(t) = p_0\, \mathbf{e}^{((b - d)\, t)}$$

```
> soln2:=rhs(%);
```

$$soln2 := p_0\, \mathbf{e}^{((b - d)\, t)}$$

```
> p[0]:=2;b:=3;d:=2;
```

$$p_0 := 2$$

$$b := 3$$

$$d := 2$$

```
> soln2;
```

$$2\, \mathbf{e}^{t}$$

```
> plot(soln2,t=0..1,thickness=2);
```



Consider the pendulum $m \frac{d^2}{dt^2} \phi + \frac{c}{l} \frac{d}{dt} \phi + \frac{m\, g}{l} \phi = 0$

```
> m:='m';c:='c';l:='l';g:=10;
```

$$m := m$$

$$c := c$$

$$l := l$$

$$g := 10$$

```
> eq3:=m*diff(phi(t),t,t)+(c/l)*diff(phi(t),t)+(m*g/l)*phi(t)=0;
```

$$eq3 := m\left(\frac{d^2}{dt^2}\phi(t)\right) + \frac{c\left(\frac{d}{dt}\phi(t)\right)}{l} + \frac{10\, m\, \phi(t)}{l} = 0$$

```
> dsolve(eq3,phi(t));
```

$$\phi(t) = \_C1\, \mathbf{e}^{\left(-\frac{\left(c+\sqrt{c^2-40\, m^2\, l}\right)t}{2\, l\, m}\right)} + \_C2\, \mathbf{e}^{\left(-\frac{\left(c-\sqrt{c^2-40\, m^2\, l}\right)t}{2\, l\, m}\right)}$$

```
> m:=1/10;c:=0;l:=4;g:=10;
```

$$m := \frac{1}{10}$$

$$c := 0$$

$$l := 4$$

$$g := 10$$

```
> eq3:=m*diff(phi(t),t,t)+(c/l)*diff(phi(t),t)+(m*g/l)*phi(t)=0;
```

$$eq3 := \frac{1}{10}\frac{d^2}{dt^2}\phi(t) + \frac{1}{4}\phi(t) = 0$$

```
> ic3:=phi(0)=1,D(phi)(0)=0;
```

$$ic3 := \phi(0) = 1, \mathrm{D}(\phi)(0) = 0$$

```
> dsolve({eq3,ic3},phi(t));
```

$$\phi(t) = \cos\left(\frac{1}{2}\sqrt{10}\, t\right)$$

```
> soln3:=rhs(%);
```

$$soln3 := \cos\left(\frac{1}{2}\sqrt{10}\, t\right)$$

```
> plot(soln3,t=0..15,thickness=2);
```




```
> p:=unapply(soln3,t);
```

$$p := t \rightarrow \cos\left(\frac{1}{2}\sqrt{10}\,t\right)$$

```
> L:=(x,y)->line([0,0],[x,y],thickness=3,color=red):
> N:=50;
```

$$N := 50$$

```
> x:=l*sin(p((15/N)*n));y:=-l*cos(p((15/N)*n));
```

$$x := 4\sin\left(\cos\left(\frac{3}{20}\sqrt{10}\,n\right)\right)$$

$$y := -4\cos\left(\cos\left(\frac{3}{20}\sqrt{10}\,n\right)\right)$$

```
> seq1:=seq(L(x,y),n=0..N):
> display(seq1,insequence=true);
```



-3
-2
-1
0
1
2
3
0
-1
-2
-3
-4